
工作草稿 — 请勿引用

KidLisp 语言参考

12个类别中的 118个生成艺术内置函数
语法、语义、时序、色彩与 $-Code组合系统

Aesthetic.Computer

[ 工作草稿 — 请勿引用]

摘要。 KidLisp 是一种用于生成艺术的极简 Lisp 方言，拥有 12 个类别中的 118 个内置函数，没有用户自定义函数、
没有递归、没有文件 I/O、没有网络访问。它在设计上就是安全的：KidLisp 程序无法突破其沙箱、无法使宿主崩溃、
也无法访问画布之外的任何资源。本文作为完整的语言参考，以语法高亮的示例记录了每个内置函数、时序系统（基于
帧和基于秒的调度，无需回调）、色彩模型（命名颜色、RGB、渐变、图案）、$-code 组合系统（通过引用嵌入程序作为
主要的抽象机制）以及求值模型（每帧对完整 AST 重新求值，使用确定性伪随机数生成器）。我们将 KidLisp 呈现为
一种围绕省略而设计的语言：它不能做什么与它能做什么同样经过深思熟虑，而这些限制产生了一个创作空间，59 位
作者已用它编写了 16,634 个程序。

1. 语法

KidLisp 使用 S-表达式语法 [McCarthy, 1960]。每个
表达式要么是原子（数字、字符串、符号），要么是用括
号括起来的列表：

; Atoms
42 ; number

"hello" ; string
red ; symbol (color name)

; Lists (function application)
(wipe "navy") ; clear screen navy
(ink "red") ; set drawing color
(box 10 10 50 50) ; draw rectangle

(circle 64 64 30) ; draw circle

没有运算符、没有中缀表示法、除了括号、字符串、数
字和符号之外没有任何特殊语法。算术使用前缀表示法：

(+ 1 2) ; => 3
(* 3 (+ 1 2)) ; => 9
(- 10 3) ; => 7
(/ 100 4) ; => 25
(% 17 5) ; => 2 (modulo)

2. 12 个类别

2.1 图形（9 个函数）

绘图原语以即时模式运行 —— 每次调用直接渲染到
帧缓冲区。

(wipe "black") ; clear canvas
(ink "cyan") ; set color
(line 0 0 128 128) ; diagonal line
(box 10 10 40 40) ; rectangle
(circle 64 64 20) ; circle at center
(tri 32 10 10 54 54 54) ; triangle
(plot 64 64) ; single pixel
(flood 64 64) ; flood fill
(shape 10 10 50 10 30 50) ; polygon

fill和 outline控制形状是实心还是描边（与 Process-
ing 兼容）：
(fill) ; filled shapes
(ink "red")
(circle 64 64 30) ; solid red circle

(outline) ; outlined shapes
(ink "white")
(circle 64 64 30) ; white ring

2.2 变换（11 个函数）

像素级变换在绘制之后作用于帧缓冲区：

(scroll 1 0) ; shift pixels right
(zoom 1.01) ; slow zoom in
(spin 1) ; rotate 1 degree
(blur 2) ; gaussian blur
(contrast 1.2) ; increase contrast
(suck 0.5) ; radial displacement
(sort) ; sort pixels by brightness

变换在帧之间累积，无需显式状态管理即可产生动
画：

(wipe "black")
(ink "white")
(circle 64 64 10) ; draw circle
(spin 2) ; rotate 2 deg/frame
(zoom 1.005) ; grow slightly/frame
; Result: spiraling , growing circle

2.3 数学（14 个函数）

(sin (* frame 0.05)) ; sine wave
(cos (* frame 0.03)) ; cosine wave
(random) ; 0-255
(random 10) ; 0-9
(random 5 15) ; 5-14
(wiggle 10) ; random +/-5
(floor 3.7) ; => 3
(abs -5) ; => 5
(sqrt 16) ; => 4
(min 3 7 1) ; => 1
(max 3 7 1) ; => 7

random 使用由程序短代码初始化的种子伪随机数生成
器，使所有生成输出具有确定性：同一个程序始终产生
相同的视觉效果。

2.4 色彩（19+ 个命名颜色）

颜色可以用名称、RGB 值或渐变字符串来指定：
(ink "red") ; named color
(ink 255 0 0) ; RGB
(ink 255 0 0 128) ; RGBA (semi-transparent)
(ink "fade:red-blue") ; horizontal gradient
(ink (fade "cyan" "magenta" "vertical"))
(ink "rainbow") ; repeating rainbow
(ink "zebra") ; striped pattern

可用的颜色名称包括所有 CSS 命名颜色：red、green、
blue、yellow、orange、purple、pink、cyan、magenta、black、
white、gray、brown、lime、navy、teal、gold、coral、salmon、
crimson、lavender、indigo、violet、turquoise、tomato 等。

在第一行使用裸颜色名称可设置背景：

"navy"
(ink "gold")
(circle 64 64 30)
; Navy background , gold circle

2.5 系统与显示（9 个函数）

(def w width) ; screen width
(def h height) ; screen height
frame ; current frame number
time ; seconds elapsed

(resolution "half") ; half-resolution
(resolution 64 64) ; explicit 64x64
(fps 30) ; set frame rate

2.6 控制流（7 个函数）

; Variable definition
(def x 50)
(def y (+ 10 (* frame 0.5)))

; Conditional
(if (> frame 60)

(ink "red")
(ink "blue"))

2



WO
RK

IN
G
DR

AF
T

工作草稿 — 请勿引用

; Loop
(repeat 10 i

(circle (* i 12) 64 5))

; Execute once (not every frame)
(once (wipe "black"))

; Local binding
(let ((cx 64) (cy 64))

(circle cx cy 30))

repeat 循环将其计数器变量（上面的 i）绑定为从 0
到 𝑛 − 1。嵌套的 repeat 使用不同的变量名：
(repeat 8 i

(repeat 8 j
(box (* i 16) (* j 16) 14 14)))

; 8x8 grid of squares

2.7 时序

KidLisp 的时序系统使用内联调度而非回调。时序表
达式包裹代码块：

; Frame -based timing
(0 (wipe "black")) ; every frame
(1 (spin 5)) ; every 2nd frame
(30f (ink "red")) ; after 30 frames

; Second -based timing
(1s (wipe "red")) ; after 1 second
(2.5s (ink "gold")) ; after 2.5 seconds
(1s! (write "!" 10 10)) ; fire once at 1s

; Repeating intervals
(0.5s... (spin 45)) ; every 0.5 seconds
(1s... (scroll 1 0)) ; every second

这消除了对 setTimeout、setInterval 或动画帧回调的
需求。时间是表达式的属性，而非对其调用的函数。

2.8 文本（4 个函数）

(write "hello" 10 20) ; text at position
(write "centered" ; centered text

(/ width 2) (/ height 2))
(type "typed" 10 10) ; typewriter effect

2.9 输入（3 个函数）

; Pen/mouse position and state
(if pen

(circle pen.x pen.y 5))

; Touch input
(if hand

(line hand.x hand.y 64 64))

2.10 音频（6 个函数）

; Microphone input
(def vol (mic)) ; amplitude 0-255

(circle 64 64 vol) ; sound -reactive

; Melody playback
(melody "C4 E4 G4 C5")

; Overtone harmonics
(overtone 440 3) ; 3rd harmonic of A4

2.11 3D 图形（8 个函数）

(cube "mycube") ; create 3D cube
(form "mycube"

(move 0 0 -5) ; position
(scale 2) ; size
(spin (* frame 0.5) 1 1 0)) ; rotate

2.12 通过 $-Code 进行组合

KidLisp 最独特的特性是其组合系统。程序不使用用
户自定义函数，而是通过短代码引用其他程序：

; Embed another program as a layer
($cow) ; execute program "cow"
($nece 10 20) ; pass parameters

; Multi -layer composition
(wipe "black")
($cow) ; background layer
($faim) ; foreground layer

; Chain multiple programs
(resolution 64 64)
($zobi)
(spin 1)
($grel)

程序从 MongoDB 获取，缓存在 RAM 和 IndexedDB
中，并通过 SHA-256 进行内容寻址。$-code 系统用单一
的社交机制取代了用户自定义函数、模块和库：找到别
人制作的程序，通过代码引用它，与之组合。

被引用最多的程序是简单的视觉原语，充当构建模块。
该系统产生了一个社交依赖图：程序 A 嵌入程序 B，程
序 B 又嵌入程序 C，在观察到的使用中深达 7 层。

3. 求值模型

KidLisp 在每一帧重新求值其整个 AST。除了显式定
义的变量外，没有保留状态。frame 和 time 变量自动递
增。zoom、scroll 和 spin 等变换在帧缓冲区中累积。

该模型消除了 Processing [Reas and Fry, 2007]中常见
的 setup()/draw()分离。KidLisp程序的每一行在每一帧
都会执行。动画源于依赖帧的表达式：

(wipe "black")
(ink "white")
; Circle orbits center using sin/cos
(circle

(+ 64 (* 30 (cos (* frame 0.03))))
(+ 64 (* 30 (sin (* frame 0.03))))
8)

3.1 确定性随机

random 的种子来自程序的短代码。同一个程序在每台
设备上、每一次都产生相同的输出。这使 KidLisp 程序
成为可复现的生成艺术：代码就是艺术品，而不仅仅是
生成它的指令。

3.2 混沌模式

无效输入——乱码、不匹配的括号、随机字符——不
会产生错误消息。相反，KidLisp 检测低识别率（< 30%
的已知词汇）和高特殊字符比率（> 50%），并渲染艺术
化的视觉输出。用户永远不会看到”SyntaxError”。他们
看到的是色彩。

4. 省略式设计

KidLisp 刻意缺少：
• 用户自定义函数：由 $-code 组合取代
• 递归：防止无限循环和栈溢出
• 文件 I/O：画布是唯一的输出
• 网络：程序无法获取外部数据
• 字符串操作：没有 concat、split、replace
• 错误消息：由混沌模式取代
• Import/require：由 $-code 取代
• 对象/类：数据仅有数字、字符串和列表
每一项省略都是设计决策。缺少用户自定义函数迫使

通过社交层进行组合。缺少错误消息意味着初学者永远
不会遇到满屏红色文字。缺少网络意味着每个程序都是
自包含且安全运行的。

这遵循了一个原则：语言的限制比其特性更能定义
其创作空间 [Papert, 1980, Compton and Mateas, 2015]。
KidLisp 不是一种碰巧能做艺术的通用语言。它是一种
不能做其他任何事情的艺术语言。

5. 完整的内置函数索引

类别 数量 核心函数

图形 9 wipe, ink, line, box, circle
变换 11 scroll, zoom, spin, blur, suck
数学 14 +, -, *, /, random, sin, cos
色彩 19+ red, blue, fade, rainbow, zebra
系统/显示 9 width, height, frame, time
控制流 7 def, if, repeat, once, let
文本/输出 4 write, type, paste, print
输入 3 pen, hand, gamepad
音频 6 mic, melody, overtone, sound
3D 图形 8 cube, form, trans, move, scale
组合 4 $code, layer, bake, embed
数据 4 list, get, set, let

合计 118

Table 1: 按类别划分的全部 118 个内置函数。

6. 示例：完整程序

一个完整的 KidLisp 程序，绘制具有轨道运动的声音
响应式粒子场：

"black"
(def vol (+ 5 (* (mic) 0.2)))
(repeat 50 i

(def angle (* i 0.1256))
(def r (+ 20 (* vol (sin (+ (* frame 0.02) angle)

))))
(def cx (+ 64 (* r (cos (+ angle (* frame 0.01)))

)))
(def cy (+ 64 (* r (sin (+ angle (* frame 0.01)))

)))
(ink (+ 100 (* i 3)) 50 (+ 150 (* i 2)))
(circle cx cy (+ 1 (random 3))))

(blur 1)
(zoom 1.002)

该程序：设置黑色背景，读取麦克风振幅，绘制 50
个在声音调制下沿轨道运动的粒子，用计算出的渐变为
它们着色，应用轻微模糊，并缓慢放大。它以 60 FPS
运行。它只有 12 行。它不需要导入、不需要设置函数、
不需要动画循环、也不需要事件监听器。

